This is to certify that the

thesis entitled

MICROPROCESSOR CONTROLLED DATA
ACQUISITION FOR RAMAN SPECTROSCOPY

presented by

RICHARD LEE BOWERSOX

has been accepted towards fulfillment
of the requirements for

MASTERS degree in PHYSICS

P. J. Colwell
Major professor

Date NOVEMBER 7 1979

O-7639

MICROPROCESSOR CONTROLLED DATA ACQUISITION FOR RAMAN SPECTROSCOPY

By

Richard Lee Bowersox

A THESIS

Submitted to
Michigan State University
in partial fulfilment of the requirements
for the degree of

MASTER OF SCIENCE

Department of Physics

1979

ABSTRACT

MICROPROCESSOR CONTROLLED DATA ACQUISITION
FOR RAMAN SPECTROSCOPY

By

Richard Lee Bowersox

While Raman spectroscopy is useful in obtaining information on the bond structure of crystals with lattice damage, the collection of data during spectroscopic measurements is a very repetitive task. It can easily be performed by an automated controller if all collection apparatus can be remotely controlled. With a computer based system, the whole cycle from the beginning of counting to final equipment shut down can be simply achieved. A system with this capability is described in this thesis.

Data obtained from silicon carbide (SiC) crystals ion implanted at a dose of 1.6×10^{21} H^+/cm^3 show a graphite-type of bond structure which returns to the normal 6H crystal structure after annealing at 500°C.

ACKNOWLEDGMENTS

I would like to thank Dr. P. Colwell for her guidance in the preparation of this thesis, and the funds that made the new controller a reality.

Mr. T. Burt and Dr. P. Signell were very helpful in allowing me to use the Project Physnet microcomputer to edit and assemble my controller code.

Finally, I would like to thank my wife for understanding and putting up with my many late evenings away from home at the lab.

TABLE OF CONTENTS

LIST OF FIGURES

CHAPTER I

INTRODUCTION

Raman spectroscopy is the inelastic scattering of light from atoms or molecules. In the case of solids, a photon strikes an atom of the crystal. At this point momentum is transferred between the lattice and the photon, creating (Stokes) or annihilating (anti-Stokes) an optical phonon. The light then exits the crystal with either a lower or higher energy, respectively.

Previous studies showed that when silicon carbide (SiC) is implanted with H^+ to doses of $2.0\text{x}10^{21}/\text{cm}^2$ or $1.6\text{x}10^{21}/\text{cm}^2$, the resultant spectra showed the formation of a graphite-type bond structure. This structure had much more intensity than that of the characteristic phonons of the host crystal. Upon annealing the crystals, the 'graphite' disappeared, and the pure crystal structure returned. We wanted to know if lower implantation doses would affect the SiC in a similar way, or if a different process occurs.

To do these studies, we required many data collection runs, and it was felt that the control system could be improved to provide a more automated scheme.

For Raman spectroscopy, we use a minimum of equipment. The essential are:

1. A monochromatic light source that can be focused to a fine spot
2. A sample holder for the crystal of interest
3. A spectrometer to resolve the wavelength of light emitted by the sample
4. A device to measure the intensity of the light after it goes through the spectrometer
5. Some sort of control to coordinate the activities of the above devices.

This thesis will concentrate on the last of the above items. What I have done is to replace a hardwired control box with a microprocessor controller system. This approach has many advantages which will soon be explained.

The current controller was designed and built by Dr. Philip Gaubis, a former graduate student. It was built in 1975 and used the current electronics of the time, which consisted of small and medium scale integrated circuits of transistor-transistor logic (TTL). The design was reasonable for the time and relatively inexpensive to implement. Unfortunately this method suffers a number of disadvantages. These problems are:

1. Lack of flexibility. The controller is designed for one function and can only perform that one thing (with a few variable parameters). To change it performance would require a major re-wiring of it circuit boards.

2. A large number of circuit packages. The great number of TTL IC's consume a good amount of current (several amperes) and generate a lot of heat. They take up six separate circuit boards (in two cases), and have a tendency, of late, to work loose from their sockets (with subsequent system failure).
3. Complex system timing. Since about six separate tasks must be performed in sequence, a fair amount of logic is needed to keep them straight.

The jobs that are to be performed are:

a) Assure that the laser is running during data acquisition*

b) Start and stop data collection sequence

   i) Start photon counter (SSR Multimode Processor)

   ii) Accept data from SSR after collection period

   iii) Transfer information to incremental recorder

   iv) Increment spectrometer wavelength setting

   v) Reset SSR and restarting counting.

c) Calculate total wavelength change from start to end.*

d) Shut off laser at end of run if desired.*

e) Shut off cooling water to laser fifteen minutes after laster shut-down.*

f) Provide various types of interface for manual control of equipment.

---

*Indicates additional function designed into new controller.

Adding the new features to the existing controller would have entailed a complete rewiring. In addition, many new circuit elements would be required. This would have overloaded existing power supplies.

The change to microprocessor control results in an order of magnitude reduction in the number of individual Integrated Circuits (ICs). This gives an equivalent reduction in power requirements. Future expansion of the system is simply a matter of adding a new interface circuit, and/or software changes.

CHAPTER II

EQUIPMENT DESCRIPTION

A. Laser Modifications

The monochromatic light source used is a Coherent Radiation Model CR-5 Argon-Ion Laser. This laser is capable of emitting several different wavelengths of light. A prism is mounted in the optical cavity to isolate the desired wavelength.

To function with the new control, a modification was made to the laser head assembly. As described in the manual[1] there is an interlock switch that shuts off all power when the cover of the head is removed. To achieve remote shut down of the laser, the circuit of Figure 2.1a was added in parallel with this switch. When the base of the transistor is pulled high, the relay closes, causing a current flow to this interlock thus shutting off the laser. The switch on the remote shut off allows the control to be disabled to allow continuous operation of the laser. This circuit derives its power from the laser itself. Since the control supplies a voltage level to the interface whenever the laser is on, the computer can monitor the laser's power status.



Figure 2.1 Circuits added for remote laser shut-down.

The laser is water cooled. When it is turned off, it is desirable to allow the water to run about fifteen minutes to cool down the power supply pass elements and the laser tube. The water must be shut off though, to prevent condensation on the outside of the tube. Control of the water is provided by the circuit of Figure 2.1b. This is the same as the laser control, except that the extra relay is used to handle the current demands of the solenoid. The switch on this control has positions for water off, computer control, and water on.

B. The Spectrometer

The spectrometer is a Spex 1406, 0.85 metre Czerny-Turner Double Monochromator. It is calibrated so that the wavelength of the transmitted light can be read in angstroms.(2) The wavelength can be resolved to 0.02 angstrom steps.

The control for the spectrometer is provided by a Compu-drive (stepper-motor driven wavelength drive). This unit is capable of directly driving the spectrometer at several incremental rates. It can be controlled by an external source, in this case by the interface board. There are three control signals that must be provided, and two that are returned by the controller as status. The primary control line is the one to indicate that an external device is taking charge of the controller. The second signal needed is one to control the direction of the drive motor. The third is a line to provide the drive pulses. The output line indicates if the motor has reached either its high or low travel limit switches.

C. Multi-Mode Processor

To measure the light exiting from the spectrometer, there is a photomultiplier tube used in a photon counting mode. The photon counting is done by an SSR model 1108 multi-mode counter. This counter can be set to count pulses (photons) for a certain period of time, or to time for a set amount of pulses. All controls for the counter can be operated remotely. In this case only those affecting the transfer of data out of the counter are used. All other settings are made on the front panel of the counter. The signals needed for data transfer from the counter are:[3]

1) Thirty-two (32) lines for the parallel transmission of eight Binary Coded Decimal (BCD) digits.
2) Two (2) handshake lines for request to transmit and data accepted.
3) A reset/start line to start the counter for the next cycle.

D. Incremental Recorder

The device used for logging data is a Kennedy Model 1600 Incremental recorder. The recorder writes data on a 7-track magnetic computer tape. This unit operates almost totally under remote control. The lines required for its functioning are:[4]

1) Six (6) lines for parallel loading of Extended BCD data
2) A ready out line to indicate that recorder is ready to accept data
3) End of record (EOR) and End of File (EOF) gap inputs
4) A gap in progress output line to indicate that the recorder is currently busy inserting an EOR or EOF gap
5) A Write/Step line to initiate a tape write of data.

The last device needed is the controller which governs the functioning of all the other units. It must sequence the flow of data from spectrometer, to counter, to recorder. It should also allow a convenient access to start and stop data acquisition. It provides a means of stopping the data collection and resetting the apparatus in cases of error in a run.

E. Microprocessor Controller

As was previously described, the current controller was fabricated in 1975 out of low and medium scale integration devices. It takes up one-half of an Ortec 'nim' bin. In addition to sequencing data collection activities, it provides the following manual controls.

1) Start or Stop data collection. (On stop, equipment finishes current cycle and stops.)

2) Set wavelength stepsize for use in stepping the spectrometer.
3) Set direction of spectrometer travel.
4) Run the spectrometer, at a low speed, in the direction of travel as previously set.
5) Step the spectrometer one 'interval' in the direction previously set.
6) Increment (change position by 0.02 Å) or drive spectrometer at adjustable speed in set direction. (The latter is used to quickly reset spectrometer position for a data run.) These two controls are located in a control box attached, by cable, to main controller.
7) Generate an EOF gap on recorder.
8) Provide a visual indication of Data Collection in progress.

A few specific drawbacks of these device controls are:

1) When, during data collection, a transfer of data from counter to recorder is needed, there is no buffer storage. This results in the halting of counting until ten digits are recorded on tape. The amount of time taken for each transfer is about 0.7 seconds (out of a usual collection interaval of twenty seconds). Over a span of several thousand data points, the additional time is considerable.

2) To set the stepsize for the spectrometer, two separate switches must be used. The first of these sets a basic stepsize of an integral multiple (from 1-10) of the spectrometer increment size of 0.02 Å. The second switch multiples this stepsize by 1, 10, or 100 to give the final stepsize. Reading the spectrometer stepsize from these switches requires the knowledge of the increment size and cannot be accomplished without a close look at their position.
3) If the EOF pushbutton is continuously depressed, the logic can 'latch up' causing the recorder to generate an unending string of EOF's. This condition requires manual intervention at the recorder controls.
4) There is no means to automatically terminate a data run at its conclusion. This requires that an otherwise automated process must be manually stopped (which may be at the end of an eight hour period).

F. Microprocessor Controller

The new computerized controller remedies the drawbacks of the previous one, and adds many refinements. First to correct the specific problems noted before:

1) The interface panel has it own onboard, 'scratchpad' memory. Within a millisecond after sensing

that the counter has data ready, the controller transfers that data to memory and resets/starts the counter. After the counter has been restarted, the controller formats and sends data to the recorder.

2) The stepsize is set on front panel thumbwhell switches with direct decimal read out. The software takes care of rounding the set size to a multiple of 0.02 Å.

3) Instead of the EOF button directly generating a file gap pulse, the EOF button status is sensed by the software which then generates a single EOF pulse.

4) The front panel has two banks of thumbwhell switches which control the total wavelength interval. The first is set to indicate the starting wavelength of the spectrum. The second is set to the desired ending wavelength. Upon starting the run, these switch banks are read and the former subtracted from the latter. The difference is stored, and decremented after each step until the run is completed. At this time (if desired) the laser and water are shut down in an orderly sequence for reasons stated before.

The new controller consists of four main components. The first is the computer card holding the microprocessor support, and software memory. The second is an interface card to buffer all data and signals from the computer. It

consists of the front panel switches and also circuitry to multiplex these switches into a format easily and quickly read by the computer. Finally there are the remote laser and water controls, previously described.

G. Computer Choice

The computer card is an SD Sales Z-80 Starter Kit. It consists of a Z-80 central processor unit, a CTC counter timer chip, a PIO peripheral input/output chip, 2048 byte (8-bit word) random access memory (RAM), two S100 bus connectors (for expansion), and various support circuits. There were two main reasons for choosing this particular computer. The first and foremost reason in that it uses the Z-80 8-bit parallel processor chip by Zilog Corporation. This chip has many unique features which made it ideal for this application.

1) It has the ability to test, set, or reset any single bit, anywhere in memory. This facility is ideal for checking and setting control lines.
2) It has decimal arthmetic capabilities which allow simple handling of data from BCD switches.
3) It has block move instructions which automatically move a block of data from one part of the memory to another. This permits quick transfer of information from peripheral device to the main system memory.
4) It has many memory addressing modes which shortens and simplifies programming.

The second reason for this board choice is its well planned auxiliary circuitry. The board contains a keypad and hexidecimal Light Emitting Diode (LED) readouts. These allow programming and verification of data. Another block allows programming of Eraseable Programmable Read Only Memories (EPROM's). This allows a software system to be put into permanent storage. This memory can be set to be accessed, on power-up or system reset, to immediately begin execution of the stored program. The PIO contains two latched 8-bit Input/Output (IO) ports. These are for external device control and data transfer. The CTC has three internal timers. These can be used, along with software delay routines, to provide long time intervals between activation of external devices. Finally, the S100 connector allow separate circuit boards to be simply plugged into the main board. Through this connection they derive power, address, status, and control signals.

H. Interface Card

The interface card plugs into the S100 connector on the computer board. This card has circuits to buffer (see Figure 2.2) and to transmit information to and from the peripheral device. These devices appear to the computer as a combination of memory and/or input/output devices. The front panel is addressed as eight consecutive memory locations and the SSR data as four. This addressing allows quick block-move instructions in the Z-80 to move data into



Figure 2.2 Block diagram of controller interface.

central memory. The control signals to and from the peripheral are individually addressable bits in four IO ports. The interface card also contains 128 bytes of scratchpad RAM used for temporary system parameter storage.

I. Front Panel

The front panel consists of several banks of switches and control logic. The switches give control of:

1) Starting wavelength
2) Ending wavelength
3) Wavelength step-size
4) Start data collection
5) Stop data collection
6) Spectrometer drive direction
7) Step spectrometer
8) Run spectrometer
9) Tape EOF generation
10) Local or remote (compu-drive) spectrometer drive control
11) System reset.

The front panel also contains a light emitting diode to indicate a run in progress. The control circuitry is designed so that the switches are multiplexed into eight 'memory' locations. This multiplexing means that only sixteen lines are needed to connect the front panel to the interface (instead of sixty).

The remote controls are the remote laser shut-off and cooling water supply valve. These two were described earlier.

CHAPTER III

PROGRAM DESIGN

A. Requirements

The design of software to run the controller was a large undertaking. Figures 3.1, 3.2, and 3.3 give a flow-chart of the tasks that need to be performed. The program must sequence all data flow between the computer and peripherals. It must be able to read data from the front panel and laser during data collection in order to stop acquisition and/or shut down the system as conditions demand.

The program was designed to execute in a linear, top down, fashion. Upon computer power-up, or reset, program execution begins. The first function is to initialize the external devices. The water valve must be opened and the laser interlock disabled. The SSR must be reset and started to display counts. Next, signals must be sent to the Kennedy, or it will run randomly. Finally, the front panel latches must be reset to allow later recognition of commands. The next thing that occurs is that the processor goes into a loop waiting for front panel input;



Figure 3.1 Initialization and front panel control flowchart.



Figure 3.2 Flow Chart of data transferal between counter and recorder.



Figure 3.3 Flow Chart of laser shutdown sequence.

1) Start
2) Run
3) Step
4) EOF
5) Local/Remote.

The program always acts upon conditions four or five. It will only accept the first three inputs if the Local/Remote switch is set in the local position (processor control of Spex). After acting on the latter four requests, control passes back to the input loop.

B. Data Collection

When the start switch is sensed, the process of data collection begins. First, the laser power is checked, and if it is off, the program jumps to the ending, water shut-off sequence. If the laser is on, all front panel switches are re-read, and their settings are moved into buffer memory. The settings of the starting and stopping wavelength switches are used to calculate the total wavelength interval. The spectrometer drive direction is then set. Next the Kennedy word count is set to zero. Finally, the data collection indicator is lit, and a reset/start pulse is sent to the SSR to begin counting.

While the counting is going on, the program sits in a loop continuously checking the SSR data ready line. Upon sensing a true condition on this line, the controller brings the Printer Busy line true. This signal commands the SSR to

move the counter data to its output bus. The processor next moves this information to buffer memory, resets the Printer Busy Line, and sends a reset/start pulse to begin another acquisition cycle. Next, the Kennedy status is checked to see if it is ready to accept data. If it is not, the program loops back and checks status until it is ready. When the Kennedy is ready, the controller writes out the data to it, bracketed by octal 45's. The end characters are used by the data reduction software (resident on MSU's CDC 6500) as a preliminary check of data validity. If the 45's are absent, the program assumes that there was a tape write error and sets that data word to zero. Next the controller increments a counter that indicates the number of words sent to the Kennedy since the last EOR. If that count reaches one hundred, an EOR request is sent to the Kennedy and the word count set to zero.

The front panel stop switch is now checked. If it is true, the data collection is halted and the indicator light extinguished. The program then loops, waiting for a start indication. Also, at this time, the laser power is checked and if absent, the program jumps to the ending routines. If the laser is running, and the stop switch hasn't been depressed, the computer decrements the total wavelength change (angstroms to go) counter by the stepsize. If there is more data to be collected, the program jumps back to wait for the SSR to transmit a new data word. When the counter goes negative, it indicates that all the data has been taken for the desired wavelength interval. The computer now jumps to the ending routines.

C. Ending Sequence

At the end of the run, the computer checks a front panel switch that indicates whether a later run on the same tape is desired. If a new run is wanted, a single EOF is generated and the program branches back to receive front panel input. If the switch is set to determine the tape, six EOF's are generated. These multiple EOF's are used by the data reduction software to indicate the last of the data files stored on the tape. The computer proceeds to shut off the laser by sending a pulse to its added auto shut off circuitry. A loop is then entered in which the processor uses the CTC, and a software routine, to time out for fifteen minutes. At the end of this time interval the water is turned off and the processor halts.

CHAPTER IV

EXPERIMENTAL RAMAN SET UP

The 4765.2 Å line (~120 mw) of the Ar-ion laser was used to induce the Raman spectrum of SiC crystals. The experimental geometry is shown in Figure 4.1. The mount held the samples in a helium atmosphere to rid the spectra of gases in the air. All spectra were taken at room temperature.

These crystals were obtained from Dr. Wm. Choyke of Westinghouse Labs, and were of the 6H, hexagonal structure with tetrahedral bonds. The two crystals observed were from the same melt. Both had been ion implanted with H^+, one to a density of $1.3 \times 10^{21}/cm^3$ and the other to $1.6 \times 10^{21}/cm^3$.

The crystals were mounted and their spectra taken. Next, we annealed the samples at 500°C in a rarefied atmosphere of Ar to prevent oxidation. They were annealed for twenty-five minutes and then cooled. The samples were remounted and another set of spectra taken.



Figure 4.1 Experimental geometry.

CHAPTER V

RESULTS OF ANNEALING ION-IMPLANTED SiC

In unimplanted crystals of SiC, the first order TO (797 cm^{-1}) and LO (975 cm^{-1}) phonon peaks appear.[5] These phonons are due to the tetrahedral C-Si bonds.[6] Our $1.3x10^{21}/cm^3$ sample shows definite peaks at these wavenumbers, and the $1.6x10^{21}/cm^3$ sample has some indication of their appearance (Figure 5.1). Our samples also show broad features around 1400 cm^{-1} and 1600 cm^{-1}. Studies were done by Tuinstra and Koenig[7] on micro-crystallites of graphite. They found two peaks at 1355 cm^{-1} and 1575 cm^{-1}. They also found that these peaks shifted to slightly higher wavenumbers as the size of the crystallites was decreased.

We believe that the broad feature at 1400 cm^{-1} and 1600 cm^{-1} are due to graphite-type (planar three-fold coordinate carbon) bonds. These bonds also occur in chemical compounds such as napthelene ($C_{10}H_8$), and amorphous carbon. Polarization studies done on our crystals agree with results obtained by Solin and Kobliska in their studies of amorphous carbon.[8] Previously it was found that at implant doses of $2.0x10^{21}/cm^3$ and higher the features at 1400 cm^{-1} and



Figure 5.1 Ion Implanted SiC before annealing.

1600 cm^{-1} were the only ones in the spectra.[9] In our studies, the samples with the $1.6x10^{21}/cm^3$ implant dose showed a dominance of this type of bond (as compared to the tetrahedral bond, evidenced by the relative intensities of LO; TO peaks and the 'graphite' peaks). We then looked at a lower dose ($1.3x10^{21}/cm^3$) and found that the 'graphite' was still there, but that the lattice did not appear to have as much damage.

After annealing the spectra show the phonons of the 6H crystal again (Figure 5.2). The graphite type bonds appear to have almost annealed away. The lower implant sample seems to anneal at the same rate as the higher one.



Figure 5.2 Raman spectrum of ion implanted SiC after annealing at 500° C.

CHAPTER VI

CONCLUSION

Previous studies showed that at ion implant doses of $2.0 \times 10^{21}/cm^3$ the resultant SiC crystal showed a graphite type structure that annealed away at temperatures of 600°C.[9] Our data indicates that at doses of $1.6 \times 10^{21}/cm^3$ and even down to $1.3 \times 10^{21}/cm^3$ the lattice is damaged in a process that is the same as at the higher dose. Low doses show no graphite type structure.[10] All three dosage levels seem to anneal to the host crystal structure in a similar manner.

APPENDIX

RAMN: A PROGRAM TO CONTROL RAMAN DATA COLLECTION
WRITTEN BY R.L. BOWERSOX FOR A MASTERS THESIS

```
START:  LD SP,STACK     ;SET UP STACK POINTER
        CALL RIOSET     ;INITIALIZE RAM-I/O CHIP
        CALL PIOSET     ;INITIALIZE PIO CHIP
        EI
LOOK:   LD A,(SWITCH)   ;GRAB SWITCH SETTINGS
        BIT 3,A         ;CHECK FOR THE DIRECTION
        PUSH AF
        LD A,(MASKA)    ;GET PIO A MASK
        RES 1,A
        OUT (PIOA),A    ;SET FORWARD
        JR NZ,LOOK1     ;IF THAT IS WANTED BRANCH
        SET 1,A         ;IF NOT CHANGE TO REVERSE
        OUT (PIOA),A
        LD (MASKA),A    ;SAVE NEW DIRECTION
LOOK1:  POP AF          ;RETRIEVE STATUS
        BIT 6,A         ;CHECK FOR REMOTE SETTING
        PUSH AF
        LD A,(MASKA)    ;GET MASK AGAIN
        SET 3,A
        OUT (PIOA),A    ;SET REMOTE
        JR NZ,LOOK2     ;BRANCH IF CORRECT
        RES 3,A
        OUT (PIOA),A    ;SET LOCAL
LOOK2:  LD (MASKA),A    ;SAVE NEW SETTING
        POP AF
        BIT 3,A         ;CHECK FOR RUN
        JR NZ,LOOK3     ;BRANCH UP IF NOT
        LD HL,MASKA     ;POINT TO SWITCH STATUS
        BIT 3,(HL)      ;CHECK FOR REMOTE
        JP Z,RUNIT      ;IF NOT EXECUTE RUN
LOOK3:  BIT 0,A         ;CHECK FOR STEP
        JR NZ,LOOK4     ;NO, GO ON
        CALL STEP       ;STEP SPEX
        JR LOOK         ;START OVER
LOOK4:  BIT 7,A         ;CHECK FOR CONTINUEOUS STEP
        JR NZ,LOOK      ;NOPE, START OVER
        CALL STEP       ;STEP SPEX
        LD A,(SWITCH)   ;GET NEW SWITCH SETTINGS
        JR LOOK4        ;KEEP IT UP TILL NO MORE RUN
PIOSET: LD A,BITMOD     ;PIO BIT MODE WORD
        OUT (ACON),A    ;SET PORT A TO BIT MODE
        LD A,PAMASK     ;BIT MASK
        OUT (ACON),A
        LD A,PAINT      ;INITIAL CONDITIONS
        LD (MASKA),A    ;SAVE FOR FUTURE USE
        OUT (PIOA),A
        LD A,BITMOD     ;SET UP B
```

```
        OUT (BCON),A
        LD A,PBMASK
        OUT (BCON),A
        LD A,PBINT      ;B INITIAL CONTITIONS
        OUT (PIOB),A    ;  (STOP - BIT 3 OUT)
        LD A,PBINT2     ;TOGGLE BIT 3
        OUT (PIOB),A
        LD A,BITMOD     ;NEW MASK, BIT 3 INPUT
        OUT (BCON),A
        LD A,PBMSK2
        OUT (BCON),A
        RET
BITMOD  EQU     0CFH
ACON    EQU     82H
BCON    EQU     83H
PIOA    EQU     80H
PIOB    EQU     81H
PAMASK  EQU     80H
PBMASK  EQU     80H
PBMSK2  EQU     88H
PAINT   EQU     0CDH
PBINT   EQU     0FDH
PBINT2  EQU     0FBH
RIOSET: LD A,AMASK      ;RAM/IO PORT A MASK
        LD (ODRA),A     ;PORT A ALL OUTPUT
        LD A,AINT       ;INITIALIZE A
        LD (RIOA),A
        LD HL,BITA0
        CALL TOGGLE     ;TOGGLE BIT 0 OF A
        INC L
        INC L
        CALL TOGGLE     ;TOGGLE BIT 2 OF A
        LD A,BMASK      ;RAM/IO PORT B MASK
        LD (ODRB),A
        LD A,BINT       ;INITIALIZE PORT B
        LD (RIOB),A
        RET
AMASK   EQU     0FFH
BMASK   EQU     0F1H
ODRA    EQU     3022H
ODRB    EQU     3023H
RIOA    EQU     3020H
RIOB    EQU     3021H
AINT    EQU     05DH
BINT    EQU     067H
BITA0   EQU     3000H
TOGGLE: LD A,(HL)       ;READ BIT
        AND A           ;SET FLAGS
        PUSH HL
```

```
        POP IX            ;GET BIT ADDRESS IN IX
        JR Z,OFF          ;JUMP IF BIT WAS ZERO
ON:     LD (HL),A         ;RESET BIT
        CALL DL25US       ;25 MICROSECOND TIMEOUT
        LD (IX+10H),A     ;SET BIT AGAIN
        RET
OFF:    LD (IX+10H),A     ;SET BIT
        CALL DL25US
        LD (HL),A         ;RESET BIT AGAIN
        RET
DL25US: PUSH HL           ;11 T CYCLES  (5.5 US)
        NOP               ;04 T CYCLES  (2.0 US)
        POP HL            ;10 T CYCLES  (5.0 US)
        RET               ;10 T CYCLES  (5.0 US)
                          ;AND CALLING  (8.5 US)
                          ;       =    (26.0 US)
DO2MS:  PUSH BC           ;SAVE REGISTER
        LD C,02H          ;THIS ROUTINE TIMES OUT FOR 2
LUPE:   LD B,45H          ;   MILLISECONDS AND RETURNS
        DJNZ LUPE
        DEC C
        JR NZ,LUPE
        RET
STEP:   LD BC,(STPMSD)    ;MOVE STEP SIZE TO BC
        LD A,C            ;LOAD MSD TO A
        AND OFH           ;MASK OUT BITS
        LD C,A            ;SAVE THEM AGAIN
SUBT:   LD A,B            ;MOVE LSD TO A
        SUB 2             ;SUBTRACT STEP SIZE (0.02 A)
        DAA               ;ADJUST DECIMAL
        LD B,A            ;SAVE IT
        LD A,C            ;MOVE MSD TO A
        SBC A,0           ;BORROW IF NECESSARY
        DAA               ;ADJUST ACC
        RET M             ;RETURN IF UNDERFLOW  (NOTE, THIS STEP
                          ; ROUNDS DOWN ODD STEP SIZE CHOICES
        LD C,A            ;SAVE MSD IF NO RETURN
        LD HL,SPXSTP      ;GET ADDRESS OF SPEX STEP BIT
        CALL TOGGLE       ;TOGGLE BIT (STEP SPEX 0.02 A)
        CALL DO2MS        ;DELAY 20 MILLISECONDS
        JR SUBT           ;DO IT AGAIN
;
;      RUNIT   THIS ROUTINE IS THE ONE CONTINUALLY EXECUTING
;              WHILE DATA COLLECTION IS GOING ON.  IT CYCLES AND
;              CHECKS FOR THE STOP SWITCH BEING SET, OVER RANGE
;              OF THE SPECTROMETER STEPPING MECHANISM, AND SSR
;              READY TO PRINT. (THIS OCCURS IN LOOP DOIT)
;
;              THIS ROUTINE ALSO TAKES IN THE FRONT PANEL SWITCHES
```

```
;               AND UNPACKS THE WAVELENGTH INTERVAL PARAMETERS.  IT
;               THEN CALCULATES THE WAVELENGTH DIFFERENCE BETWEEN
;               START AND STOP AND STORES IT IN 'STOPWL'.
;
;               WE ALSO TURN ON THE RUN LIGHT AND DISABLE THE EXTERNAL
;               SPECTROMETER DRIVE INPUT
;
;       ROUTINE CALLS:          INTRVL, TOGGLE, TOTOPE, ENDUP
;               DESTROYS:       AF, DE, HL, BC
;               CHANGES:        STOPWL, STARTL, STPMSD
;
RUNIT:          LD HL,STORFP    ;LOAD ADDRESS OF FRONT
                                   ;PANEL SWITCH STORAGE
                LD DE,STOPWL    ;LOAD ADDRESS OF ENDING WAVELENGTH
                LDI             ;MOVE 2 BYTES
                LDI
                LD A,(HL)       ;GET LS-DIGIT IN A
                AND OFOH        ;PUT ZEROS IN LOWER 4 BITS
                LD (DE),A       ;STORE REST OF STOP
                INC DE
                LD A,(HL)       ;GET BACK BYTE
                AND OFH         ;KEEP LS-DIGIT
                LD (DE),A       ;STORE IN 'STPMSD'
                INC DE
                INC HL
                LDI             ;REST OF STEP SIZE
                LDI             ;STORE NEXT BYTE IN 'STARTL'
                LDI
                LD A,(HL)
                AND OFOH
                LD (DE),A
                CALL INTRVL     ;CALCULATE WAVELENGTH INTERVAL
                LD HL,RESTAR    ;RESET/START SSR
                CALL TOGGLE
                LD (LITEON),A   ;TURN ON RUN LIGHT
                LD (EXTOFF),A   ;TURN OFF EXTERNAL DRIVE ENABLE
DOIT:           LD A,(SWITCH)
                AND 08H         ;CHECK FOR STOP
                JR Z,DOIT       ;IF STOPPED, BURN UP TIME
                LD A,(RIOB)     ;CHECK TO SEE IF WE ARE
                AND OCH         ;   AT SPEX HI OR LOW LIMIT.
                JP NZ,ENDUP     ;IF SO, END
                LD A,(PIOB)     ;CHECK FOR SSR DATA READY
                AND 02H
                CALL NZ,TOTAPE  ;IF SO GET TO TAPE WRITE ROUTINE
                LD A.(ENDFLG)   ;ARE WE DONE?
                AND A
                JR Z,DOIT       ;IF NOT, LOOP AGAIN.  ALSO IF WE
                                ;   NEVER WENT ANYWHERE ON THIS LOOP
```

```
                JP ENDUP          ;IF SO,GO TO END
;
;       MOVSPX  THIS ROUTINE IS USED DURING THE EXECUTION
;               OF PROGRAMED CONTROL OF THE SPECTROMETER.
;               IT STEPS THE SPEX THROUGH THE SELECTED
;               STEP-SIZE AND UPDATES THE CURRENT POSITION
;               HELD IN LOCATION  STPMSD
;
;       ROUTINE CALLS:            DECSUB, STEP
;               DESTROYS:         HL, DE, AF
;               CHANGES:          ENDFLG
;
;
MOVSPX:         LD HL,STPMSD+2    ;ADDRESS OF STEPSIZE LSD
                LD DE,STOPWL+2    ;ADDRESS OF WAVELENGTH INTERVAL
                LD B,3            ;NUMBER OF 'DECIMAL' BYTES
                AND A             ;CLEAR CARRY
SUBTR:          CALL DECSUB       ;CALL DECIMAL SUBTRACTION ROUTINE
                DJNZ SUBTR
                JR C,SETEND       ;IF WE GO NEGATIVE, WE WILL
                                  ;    STOP STEPPING AND SET THE
                                  ;    END OF JOB FLAG
                CALL STEP         ;ELSE STEP SPEX
                RET
SETEND:         LD A,0FFH
                LD (ENDFLG),A     ;SET END FLAG
                RET
;
;       DECSUB  THIS ROUTINE PERFORMS DECIMAL
;               SUBTRACTION ON TWO NUMBERS POINTED
;               TO BY HL AND DE.  AFTER SUBTRACTION
;               DE AND HL ARE DECREMENTED (SUBTRACTION
;               MOVES FROM HIGHER ADDRESS TO LOWER ADDRESS)
;
;       ROUTINE DESTROYS:         HL, DE, AF
;               CHANGES:          (DE)  THE RESULT IS STORED HERE
;
DECSUB:         LD A,(DE)         ;GET WORD FROM STRING 1
                SBC A,(HL)        ;SUBTRACT WORD FROM STRING 2
                DAA
                LD (DE),A         ;STORE RESULT IN STRING 1
                DEC DE
                DEC HL
                RET
;
;       INTRVL  CALCULATES THE INTERVAL OF TRAVEL
;               FOR THE SPECTROMETER.
;
;       ROUTINE CALLS:            DECSUB
```

```
;               DESTROYS:        HL, DE, AF, B
;               CHANGES:         STOPWL  (RESULT IS STORED HERE)
;
INTRVL:         LD HL,STARTL+2   ;ADDRESS OF STARTING WAVELENGTH
                LD DE,STOPWL+2   ;STOP WAVELENGTN (RESULT STORED HERE)
                LD B,3           ;3 BYTES OF NUMBERS
                AND A            ;CLEAR CARRY
SUBTR1:         CALL DECSUB      ;SUBTRACT (DECIMAL)
                DJNZ SUBTR1
                RET
;
;       TOTAPE  THIS ROUTINE TAKES THE DATA FROM
;               THE SSR AND WRITES IT TO THE KENNEDY
;               RECORDER.  AT THIS TIME WE CHECK TO
;               SEE IF AN INTER-RECORD GAP IS NEEDED
;               (AFTER EVERY 100 DATA ITEMS), AND IF
;               SO GENERATE ONE.  IF KENNEDY IS TURNED OFF
;               THE PROGRAM CAN HANG WHEN WE CALL READY
;               OR ANY OTHER ROUTINE THAT CALLS READY.
;
;       ROUTINE CALLS:           TOGGLE, PRINT, PRINT2 (PART
;                                OF PRINT), READY
;               DESTROYS:        DE, BC, AF
;               CHANGES:         EORCNT
;
TOTAPE:         LD (STBUSY),A    ;SET PR BUSY ON SSR
                LD DE,(LOSSR)    ;LOWER 4 DIGITS FROM SSR
                LD BC,(HISSR)    ;HIGH 4 DIGITS
                LD (UNBUSY),A    ;RESET PRBUSY
                LD HL,RESTAR     ;SSR RESET/START ADDRESS
                CALL TOGGLE      ;RESET IT AND START IT AGAIN
                LD A,45O         ;TAPE WORD DELIMITER
                CALL PRINT2      ;TRANSFER TO KENNEDY
                LD A,D           ;GET DIGITS OUT OF
                CALL PRINT       ;   STORAGE AND CALL UNPACKING
                LD A,E           ;   AND PRINT ROUTINE
                CALL PRINT
                LD A,B
                CALL PRINT
                LD A,C
                CALL PRINT
                LD A,45O         ;DELIMITER
                CALL PRINT2
                LD HL,EORCNT     ;ADDRESS OF TAPE WORD COUNT
                LD A,99          ;HAS IT BEEN 100 WORDS
                CP (HL)          ;   SINCE LAST EOR?
                INC (HL)         ;UPDATE WORD COUNT
                RET C            ;RETURN IF <100
                XOR A            ;RESET WORD COUNT
```

```
                LD (HL),A           ;   TO ZERO
                CALL READY          ;ARE WE READY TO PRINT?
                LD HL,IRG           ;GOT THIS FAR, MUST BE
                CALL TOGGLE         ;GENERATE AN IRG
                RET
;
;       PRINT   THIS ROUTINE ACTUALLY WRITES TO
;               THE KENNEDY.
;
;       ROUTINE CALLS:              TOGGLE, READY (POSSIBLE PLACE TO HANG)
;               DESTROYS:           AF, HL
;
PRINT:          CPL                 ;COMPLIMENT ACC  (SSR SENDS
                                    ;   DATA NEGATIVE TRUE)
                PUSH AF             ;SAVE DIGITS
                AND OFH             ;MASK OUT LOWER BITS
                CALL READY          ;CHECK TO SEE IF TAPE IS READY
                OUT (PIOA),A        ;YES, GATE OUT DATA
                LD HL,WRSTEP
                CALL TOGGLE         ;PRINT IT
                POP AF              ;GET BACK MSD
                RRCA
                RRCA                ;SHIFT INTO POSITION
                RRCA
                RRCA
                AND OFH
;       PRINT2 IS THE ENTRY POINT TO WRITE
;       THE CONTENTS OF THE A REGISTER
;       TO THE KENNEDY.
PRINT2:         CALL READY
                OUT (PIOA),A        ;WRITE IT TO TAPE
                LD HL,(WRSTEP)
                CALL TOGGLE
                RET
;
;       READY   CHECKS THE STATUS OF THE KENNEDY.
;               IF IT IS OFF, THE PROGRAM WILL HANG
;               HERE UNTIL IT IS TURNED ON AND READY TO GO.
;
;       ROUTINE DESTROYS:           AF
;
READY:          IN A,(PIOA)         ;GET READY STATUS FROM A
                AND 10000000B       ;SEE IF READY
                JR Z,READY          ;IF NOT TRY AGAIN
                IN A,(PIOB)         ;YES, GET GAP STATUS
                AND 00000001B       ;SEE IF GAP IN PROGRESS
                JR NZ,READY         ;YES, START ALL OVER
                RET
```

```
;       ENDUP   THIS ROUTINE ENDS UP A RUN.
;               IF THE SWITCH 'AUTOFF' IS TRUE, THE
;               COMPUTER WILL GENERATE 6 END OF FILES
;               ON THE KENNEDY, TURN OFF THE LASER, AND
;               15 MINUTES LATER TURN OFF THE COOLING
;               WATER TO THE LASER.  THEN IT WILL RESET
;               TO THE BEGINNING OF THE PROGRAM.
;
;               IF 'AUTOFF' IS FALSE, THEN 1 END OF FILE
;               IS GENERATED AND THE LASER AND WATER REMAIN
;               TURNED ON.  THEN CONTROL IS SENT TO A WARMSTART
;               WHERE THE FRONT PANEL IS SCANNED FOR INSTRUCTIONS.
;
;       ROUTINE CALLS:          READY AND TOGGLE
;               DESTROYS:       AF, HL, BC
;
ENDUP:          LD (LITEOF),A   ;TURN OFF RUN LIGHT
                LD A,(AUTOFF)   ;CHECK TO SEE IF WE WANT THE
                AND 4H          ;    WATER AND LASER OFF WHEN DONE
                                ;   WITH THIS RUN.  ALSO END TAPE FILE.
                JR Z,ENDUP      ;YES, GO DO IT
                LD HL,EOF       ;NO, GENERATE SINGLE END OF FILE
                CALL READY      ;    - RETURN
                CALL TOGGLE
                JP START
                LD B,6          ;SET UP FOR 6 END-OF-FILES
                LD HL,EOF
FGAP:           CALL READY      ;CHECK KENNEDY STATUS
                CALL TOGGLE     ;GENERATE END OF FILE
                DJNZ FGAP       ;REPEAT
                LD HL,LASOFF    ;LOAD ADDRESS OF LASER CONTROL
                CALL TOGGLE     ;TURN OFF LASER
                LD BC,INTVEC    ;GET INTERUPT VECTOR
                LD A.B
                LD I,A
                LD A,C
                OUT (CTCO),A    ;PUT VECTOR IN TIMER
                LD BC,6B1EH     ;NUMBER OF TIMER LOOPS NEEDED
                LD A,10110110B  ;CTC MODE CONTROL WORD
                OUT (CTCO),A
                XOR A
                OUT (CTCO),A    ;LOAD A TIME CONSTANT OF 256
                IM 2            ;USE INTERUPT MODE 2
                EI
TIME:           HALT
INTVEC:         JR TIME         ;THIS IS THE TIMING LOOP
                DEC BC
                LD A,C
                AND B           ;IS BC=0?
```

```
                JR Z,TIME2          ;YES, WE ARE THROUGH AND CAN
                                    ;   TURN OFF THE WATER
                EI                  ;NO, RE-ENABLE INTERUPTS
                RETI                ;   AND RETURN
TIME2:          LD (WATOFF),A       ;TURN OFF THE WATER NOW
                JP START            ;RETURN TO LOOKING AT CONTROL BOARD

STACK   EQU     2100H
SPXSTP  EQU     3003H
CTCO    EQU     84H
LITEOF  EQU     3001H               ;TURN RUN LIGHT OFF
RESTAR  EQU     3002H               ;RUN START SSR
WRSTEP  EQU     3004H               ;WRITE/STEP KENNEDY
EOF     EQU     3005H               ;GENERATE END-OF-FILE GAP
IRG     EQU     3006H               ;GENERATE INTER-RECORD GAP
STBUSY  EQU     3008H               ;SET SSR PRINTER BUSY FLA
EXTOFF  EQU     3010H               ;DISABLE EXTERNAL DRIVE
LITEON  EQU     3011H               ;TURN RUN LIGHT ON
LASOFF  EQU     300CH               ;TURN LASER OFF
UNBUSY  EQU     3018H               ;RESET SSR PRINTER BUSY FLAG
WATOFF  EQU     3017H               ;TURN WATER OFF
LOSSR   EQU     4000H               ;LOCATION OF SSR LOW ORDER DATA
HISSR   EQU     4002H               ;DITTO FOR HI DATA FROM SSR
AUTOFF  EQU     4012H               ;THE POSITION OF THE AUTO OFF SWITCH
SWITCH  EQU     4010H               ;LOCATION OF FRONT PANEL CONTROL SWITCHES
                ORG 3080
STORFP: DS      8
STPMSD: DS      3
STOPWL: DS      3
STARTL: DS      3
ENDFLG: DS      1
EORCNT: DS      1

        END
```

LIST OF REFERENCES

LIST OF REFERENCES

1. Coherent Radiation CR-5 Argon-Ion Laser Operation Manual, schematic C125-104(H).
2. Spex Industries Model 1406 Spectrometer Instruction Manual.
3. SSR Instrument Co., Model 1108 Multi-Mode Processor Preliminary Instruction Manual.
4. Operation and Maintenance Manual, Kennedy Model 1600 Incremental Tape Recorder.
5. P. A. Gaubis, unpublished Ph.D. dissertation, Michigan State University, East Lansing (1977).
6. P. J. Colwell, unpublished Ph.D. dissertation, University of Illinois, Urbana (1971).
7. F. Tuinstra and J. L. Koenig, J. Chem. Phys. 53, 1126 (1970).
8. S. A. Solin and R. J. Kobliska, in Amorphous and Liquid Semiconductors, Vol. 2, edited by J. Stuke and W. Brenig (John Wiley and Sons, Inc., New York, 1974), p. 46.
9. P. J. Colwell, R. L. Bowersox, J. Cunningham et al., research data on SiC, Michigan State University, East Lansing, 1976-78 (unpublished).
10. Ibid.